
Organização da aula

Retomada: LPO, fatos, regras, unificação, encadeamento reverso e retrocesso.

Programação lógica e estrutura de um programa Prolog.

Fatos, regras, consultas e variáveis.

Unificação, resolução de metas, backtracking e recursão.

Exemplo integrador: alcançabilidade em grafos e relação com busca em
profundidade.

Material complementar para estudo: aritmética, estruturas e listas.

Respostas comentadas das atividades para revisão após a aula.

Fio condutor

O que muda quando fatos, regras e mecanismos de inferência deixam de ser apenas
teoria lógica e passam a constituir um programa executável?

Retomada: de LPO para Prolog

Retomando

Na Aula 06, estudamos como representar conhecimento em LPO e como derivar
conclusões por substituição, unificação e encadeamento reverso.

fatos representam informações conhecidas;

regras representam relações condicionais;

consultas expressam metas a serem provadas;

unificação compatibiliza termos;

encadeamento reverso parte da meta e procura como prová-la;

retrocesso permite tentar alternativas quando uma escolha falha.

Pergunta

E se uma linguagem de programação adotasse exatamente esses mecanismos como
modelo de execução?

Da lógica para um programa executável

Base de conhecimento
fatos + regras

Consulta
meta

Unificação Busca por uma prova
+ retrocesso

Resposta
substituições

Ideia central

Em Prolog, programar significa descrever relações e deixar que o mecanismo de
inferência procure respostas para as consultas.

O que é Prolog?

Prolog é uma linguagem de programação lógica.

Seu núcleo trabalha com fatos e regras próximas de cláusulas de Horn, com
raízes na Lógica de Primeira Ordem.

O núcleo da linguagem possui padronização ISO; implementações concretas
adicionam extensões.

Nesta disciplina utilizaremos a sintaxe e o comportamento do SWI-Prolog.

Discussão

Em uma linguagem imperativa, normalmente descrevemos como calcular. Em Prolog,
procuramos descrever o que é verdadeiro sobre o domínio e quais relações devem
valer.

Programação declarativa e leitura operacional

Leitura declarativa

O programa descreve relações entre
objetos e condições para que essas
relações sejam verdadeiras.

Leitura operacional

O Prolog tenta satisfazer metas
escolhendo cláusulas, unificando termos
e executando submetas em uma ordem
definida.

Observação

Entender Prolog exige as duas leituras. Um programa pode estar logicamente correto
e ainda assim ter comportamento operacional ruim, repetir respostas ou não terminar.

Nosso primeiro programa: fatos

Considere uma pequena base de conhecimento sobre relações familiares:

progenitor(ana, bruno).
progenitor(bruno, clara).
progenitor(bruno, diego).
progenitor(clara, eva).

progenitor é o nome do predicado;

ana, bruno, clara etc. são átomos;

cada fato termina com ponto.

Pergunta

Qual informação cada argumento de progenitor(X,Y) representa? A ordem dos
argumentos importa?

Termos em Prolog

Os principais tipos de termos usados nesta aula são:

átomos: ana, azul, no_centro;

números: 3, 4.5;

variáveis: X, Pessoa, Resultado;

termos compostos: ponto(2,4), progenitor(ana,bruno).

Observação

Átomos normalmente começam com letra minúscula; variáveis começam com letra
maiúscula ou com sublinhado.

Consultas: perguntar à base de conhecimento

Com o programa carregado, podemos consultar:

?- progenitor(ana, bruno).
true.

?- progenitor(ana, diego).
false.

Pergunta

O segundo resultado significa que é logicamente impossível Ana ser progenitora de
Diego, ou apenas que o programa não consegue provar isso com o conhecimento
disponível?

Consultas com variáveis

Uma variável transforma a consulta em uma busca por valores que a tornam
verdadeira.

?- progenitor(bruno, X).
X = clara ;
X = diego ;
false.

a primeira resposta fornece uma substituição para X;

o ponto e vírgula solicita outra solução;

ao esgotar as alternativas, não há nova solução a retornar.

Conexão com LPO

A resposta X = clara é uma substituição encontrada por unificação durante a busca
pela prova.

Exercício 1 - fatos e consultas

Individual - 5 minutos

Considere a base a seguir. Antes de executar no Prolog, escreva a resposta esperada
para cada consulta.

progenitor(ana, bruno).
progenitor(bruno, clara).
progenitor(bruno, diego).
progenitor(clara, eva).

1 ?- progenitor(bruno, clara).

2 ?- progenitor(ana, X).

3 ?- progenitor(X, eva).

4 ?- progenitor(bruno, X).

Depois de executar

Em qual consulta o Prolog precisa usar retrocesso para apresentar todas as
respostas?

Exercício 1 - resposta comentada

Respostas esperadas

Cada consulta é resolvida procurando fatos que unifiquem com a meta.

?- progenitor(bruno, clara).
true.

?- progenitor(ana, X).
X = bruno.

?- progenitor(X, eva).
X = clara.

?- progenitor(bruno, X).
X = clara ;
X = diego ;
false.

Onde aparece o retrocesso?

Na quarta consulta existem duas respostas. Depois de X = clara, o ponto e vírgula
pede outra solução e o Prolog retrocede até a escolha do fato
progenitor(bruno,X) para tentar o próximo fato compatível. 11 / 83

De consultas compostas para regras

Podemos perguntar quem é avô ou avó de Eva sem ter criado ainda um predicado
específico:

?- progenitor(X, Y), progenitor(Y, eva).
X = bruno,
Y = clara.

a vírgula representa conjunção de metas;

uma mesma variável em duas metas deve receber valores compatíveis;

o resultado intermediário Y = clara conecta as duas relações.

Definindo relações por regras

Em vez de repetir a consulta composta, podemos definir uma nova relação:

avo_ou_avoa(X, Z) :-
progenitor(X, Y),
progenitor(Y, Z).

Uma leitura possível é:
“X é avô ou avó de Z se existe Y tal que X é progenitor de Y e Y é progenitor
de Z.”

Observação

Em Cabeca :- Corpo., a cabeça é verdadeira quando o corpo pode ser provado.

Fatos, regras e consultas

Elemento Papel Exemplo

fato conhecimento incondici-
onal

progenitor(ana,bruno).

regra relação condicional ancestral(X,Z) :- ...
consulta meta que queremos pro-

var
?- ancestral(ana,X).

Cheque sua compreensão

Em qual desses três elementos aparecem as respostas que o Prolog devolve ao
usuário?

Unificação: o mecanismo de encaixe

Retomando

Na Aula 06, unificar significava encontrar uma substituição que tornasse dois termos
idênticos.

Exemplos:
progenitor(X , bruno) e progenitor(ana, bruno)

unificam com
{X/ana}.

Já
progenitor(X , bruno) e progenitor(ana, clara)

não unificam, pois os segundos argumentos são incompatíveis.

15 / 83



Centro Federal de Educação Tecnológica de Minas Gerais - CEFET-MG, campus Divinópolis - Prof. Tiago Alves de Oliveira

Unificação diretamente no Prolog

O operador = tenta unificar dois termos:

?- X = ana.
X = ana.

?- pessoa(X, 20) = pessoa(carla, Idade).
X = carla,
Idade = 20.

?- ponto(X, 2) = ponto(3, 4).
false.

Observação

= não significa “atribuição” como em muitas linguagens imperativas. Seu papel é
unificação.

16 / 83



Centro Federal de Educação Tecnológica de Minas Gerais - CEFET-MG, campus Divinópolis - Prof. Tiago Alves de Oliveira

Exercício 2 - escrevendo regras

Em duplas - 7 minutos

Usando somente os fatos progenitor/2, escreva regras para as relações abaixo.

1 filho(Filho, Progenitor) - relação inversa de progenitor/2;
2 irmaos(X,Y) - X e Y possuem um progenitor em comum;
3 neto(Neto, Avo) - existe uma pessoa intermediária entre avô e neto.

Teste mental

A regra de irmãos também permite concluir irmaos(clara,clara)? Se sim, como
você impediria essa resposta?

Exercício 2 - resposta comentada

1. Relação inversa: filho/2

filho(Filho, Progenitor) :-
progenitor(Progenitor, Filho).

2. Irmãos: um progenitor em comum

irmaos(X, Y) :-
progenitor(P, X),
progenitor(P, Y),
X \= Y.

3. Neto: uma pessoa intermediária

neto(Neto, Avo) :-
progenitor(Avo, P),
progenitor(P, Neto).

Por que precisamos de X ̸= Y?

Sem a restrição, ter um progenitor em comum é suficiente:

irmaos(X, Y) :- progenitor(P, X), progenitor(P, Y).

Considere a consulta:

?- irmaos(clara, clara).

Se a KB contém progenitor(bruno, clara)., as duas submetas tornam-se
exatamente o mesmo fato:

progenitor(bruno, clara), progenitor(bruno, clara).

Consequência

A consulta teria sucesso. Para excluir esse caso, declaramos explicitamente X \= Y.
Prolog executa o que foi formalizado; ele não acrescenta a interpretação intuitiva de
“irmãos”.

Da intenção à regra formal

A primeira versão expressava apenas:

∃P
(
Progenitor(P,X) ∧ Progenitor(P,Y )

)
.

Isso é verdadeiro inclusive quando X = Y .
Para representar a relação desejada, acrescentamos a condição:

X ̸= Y .

Assim, a ideia completa fica:

Irmaos(X ,Y )← Progenitor(P,X) ∧ Progenitor(P,Y ) ∧ X ̸= Y .

Mensagem central

Prolog não completa o significado que pretendemos. Ele executa exatamente as
condições que formalizamos na regra.

Exercício 3 - unificação

Individual - 5 minutos

Para cada par, determine se há unificação. Quando houver, escreva a substituição
mais geral.

1 pessoa(X,20) e pessoa(ana,Y)

2 ponto(X,2) e ponto(3,4)

3 f(X,g(Y)) e f(a,g(b))

4 progenitor(X,X) e progenitor(ana,bruno)

Pergunta

Qual restrição impede a unificação do último par?

Exercício 3 - resposta comentada

1 pessoa(X,20) e pessoa(ana,Y)
Unificam com X = ana e Y = 20.

2 ponto(X,2) e ponto(3,4)
Não unificam: o segundo argumento exigiria 2 = 4.

3 f(X,g(Y)) e f(a,g(b))
Unificam com X = a e Y = b.

4 progenitor(X,X) e progenitor(ana,bruno)
Não unificam: a mesma variável X teria de ser simultaneamente ana e bruno.

Ideia central

Uma mesma variável, dentro do mesmo termo ou cláusula, deve manter uma
substituição consistente em todas as suas ocorrências.

Como o Prolog tenta provar uma meta?

Para uma consulta, o Prolog executa essencialmente este ciclo:
1 toma a primeira meta ainda não resolvida;
2 procura, de cima para baixo, uma cláusula cuja cabeça unifique com essa meta;
3 aplica a substituição obtida pela unificação;
4 se a cláusula for uma regra, substitui a meta pelas submetas do corpo;
5 tenta provar as submetas da esquerda para a direita;
6 se houver falha, retorna ao ponto de escolha mais recente e tenta outra cláusula.

Conexão

É a materialização operacional do encadeamento reverso estudado na Aula 06.

Prova direta e prova por regras

Nesta aula, usaremos dois nomes didáticos para acompanhar a execução:

Prova direta por fato

A meta unifica diretamente com um fato
da base de conhecimento.

Prova por regra

A meta unifica com a cabeça de uma
regra e é substituída pelas submetas de
seu corpo.

Não confundir

“Direta” aqui significa resolver a meta com um fato. O mecanismo de execução do
Prolog continua sendo orientado pela meta, isto é, essencialmente encadeamento
reverso — não encadeamento direto.

Prova direta por fato - passo a passo

Considere a consulta:

?- progenitor(bruno, clara).

1 Meta atual: progenitor(bruno,clara).
2 O Prolog percorre as cláusulas de progenitor/2 na ordem em que aparecem.
3 progenitor(ana,bruno) não unifica com a meta.
4 progenitor(bruno,clara) unifica sem necessidade de substituir variáveis.
5 Como o fato não possui corpo, não surgem novas submetas.
6 A consulta termina com true.

O que houve de inferência?

Mesmo neste caso simples, houve seleção de cláusula e tentativa de unificação.
Apenas não foi necessário expandir nenhuma regra.

Prova direta com variável - de onde vem a resposta?

Agora consulte:

?- progenitor(bruno, X).

1 a meta contém a variável livre X;
2 o fato progenitor(bruno,clara) unifica com a meta;
3 a unificação produz a substituição X = clara;
4 o Prolog devolve essa substituição como primeira resposta;
5 existe ainda outra cláusula compatível: progenitor(bruno,diego).

Ponto de escolha

Como existe mais de uma cláusula possível, o Prolog guarda informação suficiente
para retornar e procurar outra resposta caso o usuário digite ;.

Retrocesso na prova direta - passo a passo

Depois da primeira resposta:

X = clara ;

1 o usuário solicita outra solução com ;;
2 o Prolog retorna ao último ponto de escolha;
3 a substituição X = clara é desfeita;
4 a próxima cláusula compatível é testada;
5 progenitor(bruno,diego) gera X = diego;
6 se outra solução for pedida e não houver mais fatos compatíveis, a busca

termina.

Ligação com Busca

O retrocesso desfaz uma escolha e continua a exploração a partir da alternativa
seguinte, como em uma busca em profundidade com backtracking.

Prova por regra - base usada no exemplo

Considere novamente:

progenitor(ana, bruno).
progenitor(bruno, clara).
progenitor(bruno, diego).
progenitor(clara, eva).

avo_ou_avoa(X, Z) :-
progenitor(X, Y),
progenitor(Y, Z).

E a consulta:

?- avo_ou_avoa(bruno, eva).

Antes de executar

Qual variável intermediária precisa ser descoberta para conectar bruno a eva?

Prova por regra - passo 1: escolher e unificar a regra

Meta inicial:
avo_ou_avoa(bruno, eva)

A cabeça da regra é:
avo_ou_avoa(X ,Z )

A unificação produz:
{X/bruno, Z/eva}.

Aplicando essa substituição ao corpo, a meta original é substituída por:

progenitor(bruno,Y ), progenitor(Y , eva).

Observação

A consulta ainda não foi provada. A regra apenas transformou uma meta em duas
submetas.

Prova por regra - passo 2: resolver a primeira submeta

Primeira submeta, pois o Prolog trabalha da esquerda para a direita:

progenitor(bruno,Y )

O primeiro fato compatível é:

progenitor(bruno, clara).

Logo:
{Y/clara}.

A segunda submeta passa a ser:

progenitor(clara, eva).

Ponto de escolha

Como também existe progenitor(bruno,diego), fica registrado um ponto de
escolha para Y.

Prova por regra - passo 3: resolver a segunda submeta

Agora é necessário provar:
progenitor(clara, eva).

Esse fato existe diretamente na base de conhecimento. Portanto:

progenitor(clara, eva) ✓

e todas as submetas do corpo foram satisfeitas.

Conclusão

A consulta avo_ou_avoa(bruno,eva) tem sucesso porque foi construída uma
prova completa a partir da regra e dos fatos.

Registro da prova por regra

Etapa Meta escolhida Resultado

1 avo_ou_avoa(bruno,eva) regra selecionada; X=bruno,
Z=eva

2 progenitor(bruno,Y) fato selecionado; Y=clara
3 progenitor(clara,eva) fato encontrado; sucesso

Retomando

A execução pode ser acompanhada como uma sequência repetida de meta→
cláusula→ unificação→ novas metas.

E quando pedimos outra solução?

Considere agora uma consulta que deixa a resposta aberta:

?- avo_ou_avoa(bruno, Z).

1 primeira escolha: Y = clara;
2 a submeta progenitor(clara,Z) produz Z = eva;
3 ao digitar ;, o Prolog volta ao ponto de escolha de progenitor(bruno,Y);
4 desfaz Y = clara e tenta Y = diego;
5 passa a procurar progenitor(diego,Z);
6 não há fato compatível: esse ramo falha e não surgem novas respostas.

Árvore de prova e pontos de escolha

avo_ou_avoa(bruno,Z)

Y=clara

progenitor(clara,Z)
Z=eva ✓

Y=diego

progenitor(diego,Z)
falha

Observação

O ramo da esquerda produz uma resposta. O ramo da direita só é explorado quando
se pede outra solução ou quando a primeira alternativa não permite concluir a
consulta.

Retrocesso - backtracking

O ciclo operacional do retrocesso pode ser guardado assim:
1 escolher uma cláusula ou resposta possível;
2 unificar e propagar as substituições;
3 avançar para as metas seguintes;
4 se houver falha, desfazer substituições até o último ponto de escolha;
5 tentar a próxima alternativa.

Modelo mental

Que informação precisa ser preservada em um ponto de escolha para que a execução
possa ser retomada corretamente?

A ordem do programa pode importar

Considere duas versões de uma regra recursiva:

ancestral(X, Z) :- progenitor(X, Z).
ancestral(X, Z) :- progenitor(X, Y), ancestral(Y, Z).

A ordem das cláusulas e das metas influencia a ordem de exploração.

Consequência

Em Prolog, equivalência lógica não garante o mesmo comportamento operacional.
Uma formulação ruim pode levar a trabalho desnecessário ou até não terminar.

Recursão: definindo ancestral

ancestral(X, Z) :- progenitor(X, Z).
ancestral(X, Z) :-

progenitor(X, Y),
ancestral(Y, Z).

primeira cláusula: caso base;

segunda cláusula: caso recursivo;

a relação pode percorrer uma cadeia de tamanho arbitrário.

Pergunta

Qual é a condição que impede a recursão de continuar indefinidamente neste
exemplo?

Trace conceitual de uma consulta recursiva

Para ancestral(ana,eva):
1 tenta provar progenitor(ana,eva) — falha;
2 tenta progenitor(ana,Y) — encontra Y=bruno;
3 passa a provar ancestral(bruno,eva);
4 encontra Y=clara;
5 passa a provar ancestral(clara,eva);
6 o caso base progenitor(clara,eva) tem sucesso.

Retomando

O programa não “calcula” explicitamente a cadeia; a cadeia emerge da combinação
entre regra recursiva, unificação e busca.

Exercício 4 - trace de recursão

Em duplas - 6 minutos

Sem executar, faça um trace simplificado da consulta abaixo, indicando as metas
geradas até chegar ao caso base.

?- ancestral(ana, eva).

1 Qual é a primeira cláusula testada?
2 Quais valores são atribuídos às variáveis intermediárias?
3 Em que ponto a recursão termina?
4 Haveria outra solução se pedíssemos mais respostas com ;?

Exercício 4 - resposta comentada

Para ?- ancestral(ana,eva)., o Prolog tenta primeiro o caso base e depois o
caso recursivo em cada nível.

1 progenitor(ana,eva) falha.
2 Regra recursiva: progenitor(ana,Y) produz Y=bruno.
3 Nova meta: ancestral(bruno,eva).
4 O caso base progenitor(bruno,eva) falha; na recursão, primeiro surge

Y=clara.
5 Nova meta: ancestral(clara,eva).
6 O caso base progenitor(clara,eva) tem sucesso.

Substituições intermediárias

A cadeia relevante é ana -> bruno -> clara -> eva. A prova termina quando o
caso base encontra o fato progenitor(clara,eva).

Exercício 4 - e se pedirmos outra resposta?

A consulta é totalmente instanciada:

?- ancestral(ana, eva).

Ela pergunta apenas se existe uma prova. Na base atual há uma única cadeia que
demonstra essa relação.

ao solicitar outra solução com ;, o Prolog volta aos pontos de escolha
anteriores;

tenta, por exemplo, a alternativa progenitor(bruno,diego);

como diego não leva a eva, essa alternativa falha;

não restando outra prova, a busca termina.

Observação

Consultas com variáveis podem produzir novas substituições. Nesta consulta fechada,
uma prova alternativa apenas confirmaria novamente o mesmo resultado lógico:
sucesso.

41 / 83



Centro Federal de Educação Tecnológica de Minas Gerais - CEFET-MG, campus Divinópolis - Prof. Tiago Alves de Oliveira

Falha não é o mesmo que falsidade clássica

quando uma consulta falha, o Prolog informa que não encontrou uma prova
com o programa disponível;

isso não é automaticamente equivalente a demonstrar a negação lógica da
consulta;

muitos programas Prolog adotam uma visão de mundo fechado: o que não é
provado é tratado operacionalmente como falha.

Cuidado conceitual

Programação lógica e lógica clássica estão relacionadas, mas o modelo operacional
do Prolog introduz escolhas e hipóteses que precisam ser compreendidas
separadamente.

Negação como falha

No Prolog padrão, \+ Meta tem sucesso quando Meta não pode ser provada.

tem_filho(X) :- progenitor(X, _).

sem_filho(X) :- pessoa(X), \+ tem_filho(X).

Observação

\+ representa negação como falha (negation as failure), e não a negação clássica da
LPO.

Pergunta

Por que consultar \+ tem_filho(X) com X livre pode ter um significado diferente
daquele que intuitivamente esperamos?

Variáveis e variável anônima

tem_filho(X) :- progenitor(X, _).

X interessa porque queremos saber quem tem filho;

_ representa uma variável cujo valor não precisamos recuperar;

cada ocorrência isolada de _ deve ser entendida como uma nova variável
anônima.

Interpretação

O que a regra afirma sobre o nome do filho? Nada: apenas exige que algum segundo
argumento exista em um fato compatível.

Conjunção e disjunção

Os principais conectivos operacionais desta aula são:

% Conjuncao: ambas as metas devem ter sucesso
mae(X, Y) :- progenitor(X, Y), mulher(X).

% Disjuncao: uma das alternativas deve ter sucesso
responsavel(X, Y) :- mae(X, Y) ; pai(X, Y).

, corresponde a uma sequência conjuntiva de metas;

; representa alternativas disjuntivas;

parênteses ajudam a tornar a precedência explícita quando combinamos
operadores.

Atividade curta - leia antes de executar

Em duplas - 6 minutos

Considere a base progenitor usada na aula e as regras avo_ou_avoa/2 e
ancestral/2. Sem executar o programa, determine a sequência provável de
respostas.

1 ?- progenitor(bruno,X).

2 ?- avo_ou_avoa(X,eva).

3 ?- ancestral(ana,X).

4 ?- ancestral(diego,X).

Pergunta

Em quais consultas existem pontos de escolha? Em quais delas o retrocesso é
necessário para obter todas as respostas?

Atividade curta - respostas esperadas

Com a ordem dos fatos e regras usada na aula:

?- progenitor(bruno, X).
X = clara ;
X = diego ;
false.

?- avo_ou_avoa(X, eva).
X = bruno ;
false.

?- ancestral(ana, X).
X = bruno ;
X = clara ;
X = diego ;
X = eva ;
false.

?- ancestral(diego, X).
false.

Atividade curta - onde aparecem os pontos de escolha?

progenitor(bruno,X) possui duas correspondências diretas: clara e
diego.

avo_ou_avoa(X,eva) encontra apenas uma resposta final, mas a busca testa
alternativas internas até construir a prova.

ancestral(ana,X) combina duas cláusulas para ancestral/2 e vários
fatos progenitor/2; o retrocesso é essencial para enumerar todos os
descendentes.

ancestral(diego,X) termina em falha após esgotar as cláusulas e fatos
compatíveis.

Leitura operacional

Um ponto de escolha apenas registra uma alternativa ainda disponível. Ele não
garante que essa alternativa produzirá outra resposta.

Conectando Prolog aos grafos estudados anteriormente

Podemos representar um grafo dirigido como fatos:

aresta(a, b).
aresta(a, c).
aresta(b, d).
aresta(c, d).
aresta(d, e).

E definir alcançabilidade:

alcanca(X, Y) :- aresta(X, Y).
alcanca(X, Y) :- aresta(X, Z), alcanca(Z, Y).

Busca como inferência

Uma consulta sobre alcançabilidade passa a ser resolvida como uma busca por uma
prova construída a partir das arestas disponíveis.

Explorando o grafo - qual regra é tentada primeiro?

Para a consulta alcanca(a,e), temos duas regras:

alcanca(X ,Y )← aresta(X ,Y )
alcanca(X ,Y )← aresta(X ,Z ), alcanca(Z ,Y )

1 primeiro, tenta provar diretamente aresta(a,e);
2 como esse fato não existe, a primeira regra falha;
3 o Prolog retrocede e tenta a regra recursiva;
4 surge a submeta aresta(a,Z) seguida de alcanca(Z,e).

Retomando

A regra direta funciona como caso base; a recursiva expande a busca para um vértice
sucessor.

Explorando o grafo - passo a passo até a primeira prova

Com a ordem dos fatos da base, a primeira exploração é:
1 aresta(a,Z) produz Z=b;
2 nova meta: alcanca(b,e);
3 tenta aresta(b,e) — falha;
4 usa a regra recursiva e aresta(b,Z) produz Z=d;
5 nova meta: alcanca(d,e);
6 o fato aresta(d,e) existe — sucesso.

Primeira prova encontrada

a → b → d → e

Explorando outra alternativa no grafo

Se outra prova for solicitada, o Prolog volta ao ponto de escolha mais recente com
alternativa disponível.

1 retorna à escolha feita em aresta(a,Z);
2 tenta a próxima aresta saindo de a: Z=c;
3 prova alcanca(c,e);
4 usa aresta(c,d);
5 por fim usa aresta(d,e).

Outra prova

a → c → d → e

Observação

A ordem dos fatos e das regras determina qual prova será encontrada primeiro.

Busca em profundidade aparecendo dentro do Prolog

No exemplo do grafo, a execução pode ser lida como:

escolher sucessor→ aprofundar→ falhar/sucesso→ retroceder

o Prolog aprofunda a primeira alternativa antes de explorar irmãos;

pontos de escolha registram alternativas ainda não utilizadas;

falha provoca retorno ao ponto de escolha mais recente;

em grafos com ciclos, uma regra ingênua pode entrar em recursão sem fim.

Pergunta

O que falta nesta implementação em comparação com uma DFS que mantém
explicitamente um conjunto de estados visitados?

Prolog também deixa decisões para o programador

Embora o mecanismo de busca seja implícito, o programador controla indiretamente a
execução por meio de:

ordem das cláusulas;

ordem das metas no corpo das regras;

escolha de representações;

condições de parada de regras recursivas;

uso consciente de operações não puramente lógicas.

Discussão

Declarativo não significa “sem algoritmo”. Parte do algoritmo está no mecanismo da
linguagem; outra parte aparece na forma como descrevemos o conhecimento.

SWI-Prolog e SWISH

Para os exemplos da disciplina, duas opções são suficientes:

SWI-Prolog: implementação completa para execução local;

SWISH: ambiente web baseado em SWI-Prolog para editar e executar
programas no navegador.

Observação

O foco da aula não é configuração de IDE ou integração com C++/Java. O objetivo é
compreender o modelo lógico e operacional da linguagem.

Erros conceituais frequentes

interpretar = como atribuição em vez de unificação;

esquecer que variáveis começam com letra maiúscula ou sublinhado;

interpretar falha de prova como negação lógica completa;

ignorar que a ordem de metas e cláusulas afeta a execução;

criar recursão sem condição de parada efetiva;

esquecer que uma nova resposta pode exigir retrocesso até um ponto de
escolha anterior.

56 / 83



Centro Federal de Educação Tecnológica de Minas Gerais - CEFET-MG, campus Divinópolis - Prof. Tiago Alves de Oliveira

Exercício - faça a exploração manualmente

Em duplas - 8 minutos

Sem executar o Prolog, registre a sequência de metas, substituições, pontos de
escolha e retrocessos da consulta abaixo.

?- avo_ou_avoa(bruno, Z).

Use uma tabela com quatro colunas:

passo meta substituição resultado

1
2
3

Depois

Em que ponto a segunda alternativa Y=diego passa a ser considerada?

Exploração manual - resposta passo a passo

Consulta: ?- avo_ou_avoa(bruno,Z).

1 A meta unifica com a cabeça da regra avo_ou_avoa(X,Z), produzindo
X=bruno.

2 Primeira submeta: progenitor(bruno,Y). O primeiro fato compatível produz
Y=clara.

3 Segunda submeta: progenitor(clara,Z). O fato compatível produz Z=eva.
4 Todas as submetas foram provadas; a resposta Z=eva é devolvida.

Primeira prova

A prova construída é bruno -> clara -> eva. Cada seta corresponde a um fato
progenitor/2 usado na demonstração.

Ponto de escolha

Ao provar progenitor(bruno,Y), o Prolog também encontra Y=diego. Essa
alternativa fica registrada para eventual retrocesso.

Exploração manual - a segunda alternativa

Ao pedir outra solução com ;, o Prolog retorna ao ponto de escolha mais recente com
alternativa ainda não usada.

1 A meta progenitor(bruno,Y) tinha duas soluções possíveis.
2 A primeira, Y=clara, já foi utilizada.
3 O retrocesso escolhe agora Y=diego.
4 A próxima meta torna-se progenitor(diego,Z).
5 Como não existe fato compatível, essa alternativa falha.
6 Não há outra escolha para Y; a busca termina.

Z = eva ;
false.

Resposta à pergunta da atividade

A alternativa Y=diego só é considerada depois que a primeira prova com Y=clara foi
devolvida e o usuário solicitou outra solução.

Atividade de fechamento - pequena base de conhecimento

Em duplas ou trios - 10 minutos

Modele em Prolog uma pequena rede de disciplinas usando fatos
prerequisito(A,B), significando que A é pré-requisito direto de B.

1 crie pelo menos seis fatos;
2 defina prerequisito_indireto(A,B) recursivamente;
3 escreva uma consulta para descobrir todos os pré-requisitos de uma disciplina;
4 identifique onde ocorre unificação e onde pode ocorrer retrocesso;
5 discuta o que acontece caso a rede contenha um ciclo.

Atividade de fechamento - uma solução possível

Há várias bases corretas. Um exemplo:

prerequisito(calculo1, calculo2).
prerequisito(calculo2, metodos_numericos).
prerequisito(programacao1, estruturas_dados).
prerequisito(estruturas_dados, inteligencia_artificial).
prerequisito(matematica_discreta, inteligencia_artificial).
prerequisito(inteligencia_artificial, pfc).

Para representar pré-requisito indireto, exigimos pelo menos uma disciplina
intermediária:

prerequisito_indireto(A, B) :-
prerequisito(A, X),
prerequisito(X, B).

prerequisito_indireto(A, B) :-
prerequisito(A, X),
prerequisito_indireto(X, B).

Atividade de fechamento - consultas e interpretação

Para obter pré-requisitos diretos e indiretos de pfc:

?- prerequisito(X, pfc) ;
prerequisito_indireto(X, pfc).

Na base de exemplo, a busca encontra relações como:

inteligencia_artificial — direto;

estruturas_dados e matematica_discreta — indiretos;

programacao1 — indireto por uma cadeia maior.

Onde aparecem os mecanismos?

A unificação associa A, B e X aos fatos; a recursão percorre cadeias; o retrocesso
testa outras disciplinas intermediárias.

E se houver um ciclo?

A recursão pode revisitar as mesmas disciplinas e não terminar. Para evitar isso,
registramos estados já visitados, como em busca em grafos.

Da teoria lógica ao comportamento do Prolog

fatos regras consulta

unificação encadeamento
reverso

retrocesso

respostas
substituições

Observação

Prolog conecta diretamente conteúdos que estudamos separadamente: representação
do conhecimento, inferência, unificação e busca.

Síntese do núcleo da aula

Prolog organiza programas em fatos, regras e consultas;

unificação transforma metas genéricas em casos concretos;

a execução tradicional combina encadeamento reverso, busca em profundidade
e retrocesso;

pontos de escolha preservam alternativas que podem ser retomadas após falha
ou solicitação de outra resposta;

recursão permite expressar relações de tamanho arbitrário;

o exemplo de grafos mostra que uma consulta pode ser entendida como busca
por uma prova.

Cheque sua compreensão

Qual é a diferença entre descrever uma relação em Prolog e escrever explicitamente
um algoritmo imperativo para percorrê-la?

Até aqui: núcleo da Aula 07

Fatos → regras → consultas → unificação →
prova → retrocesso → recursão → busca

Ponto de parada para o encontro de 100 minutos

Até este ponto estão os mecanismos centrais de Prolog para a disciplina. O restante
do material foi mantido no PDF como referência para estudo e apoio à Lista 05.

Antes de avançar

Você consegue explicar, sem executar o programa, como uma consulta gera
submetas, substituições, pontos de escolha e retrocessos?

Estudo complementar: aritmética, estruturas e listas

Como usar esta parte

Estes slides não precisam ser trabalhados integralmente no encontro de 100 minutos.
Eles ficam como material de consulta para estudo autônomo e para os exercícios
posteriores.

avaliação aritmética e diferença entre = e is;

comparação de expressões numéricas;

termos compostos como estruturas de dados;

listas, decomposição em cabeça/cauda e recursão.

Objetivo do estudo

Como os mesmos mecanismos de unificação e recursão continuam funcionando
quando passamos de relações lógicas para números e listas?

Unificação não é avaliação aritmética

Compare:

?- X = 1 + 2.
X = 1+2.

?- X is 1 + 2.
X = 3.

= unifica termos;

is avalia a expressão aritmética à direita e unifica o resultado com a esquerda.

Observação

O termo 1+2 é uma estrutura sintática válida; ele não é automaticamente reduzido ao
número 3.

Três comparações que parecem iguais, mas não são

?- 1 + 2 = 3.
false.

?- 1 + 2 == 1 + 2.
true.

?- 1 + 2 =:= 3.
true.

= tenta unificação;

== testa identidade entre termos, sem instanciar variáveis;

=:= avalia expressões e compara seus valores numéricos.

Estruturas: dados também são termos

Um termo composto pode representar uma estrutura de dados:

data(22, setembro, 2026)

ponto(3, 4)

estado(posicao(macaco, porta), caixa(janela), sem_banana)

Observação

A mesma estrutura de termos usada para representar conhecimento também serve
para organizar dados manipulados pelo programa.

Listas: uma estrutura recursiva

[a, b, c]
[]
[Cabeca | Cauda]

Para a lista [ana, bruno, clara]:

cabeça: ana;

cauda: [bruno, clara].

Retomando

Essa decomposição prepara naturalmente definições recursivas sobre listas.

Exemplo clássico: pertencer a uma lista

pertence(X, [X | _]).
pertence(X, [_ | Cauda]) :-

pertence(X, Cauda).

Consultas possíveis:

?- pertence(b, [a,b,c]).
true.

?- pertence(X, [a,b,c]).
X = a ; X = b ; X = c ; false.

Pergunta

Onde estão o caso base e o passo recursivo nessa definição?

71 / 83



Centro Federal de Educação Tecnológica de Minas Gerais - CEFET-MG, campus Divinópolis - Prof. Tiago Alves de Oliveira

Exercício 5 - listas

Individual - 6 minutos

Complete os predicados abaixo usando decomposição em cabeça e cauda.

% ultimo(Elemento, Lista)
ultimo(__________________________).
ultimo(__________________________) :-

______________________________.

% tamanho(Lista, N)
tamanho([], 0).
tamanho([_|Cauda], N) :-

______________________________.

Pergunta

Qual dos dois predicados exige uma operação aritmética com is?

Exercício 5 - resposta comentada

% ultimo(Elemento, Lista)
ultimo(Elemento, [Elemento]).
ultimo(Elemento, [_|Cauda]) :-

ultimo(Elemento, Cauda).

% tamanho(Lista, N)
tamanho([], 0).
tamanho([_|Cauda], N) :-

tamanho(Cauda, N1),
N is N1 + 1.

em ultimo/2, o caso base é uma lista com um único elemento;

o passo recursivo descarta a cabeça até chegar a esse caso;

em tamanho/2, primeiro calculamos o tamanho da cauda;

N is N1 + 1 avalia a expressão aritmética e produz o tamanho da lista atual.

Resposta

tamanho/2 é o predicado que exige avaliação aritmética com is.

Exercícios de estudo complementar

Prática individual

Use os slides seguintes e a documentação do SWI-Prolog para ampliar o programa da
aula. Registre pelo menos duas consultas de teste para cada predicado implementado.

1 implemente dobro(X,Y) usando avaliação aritmética;
2 implemente primeiro/2, ultimo/2 e tamanho/2 para listas;
3 implemente concatena/3 sem usar append/3;
4 estenda a base familiar e implemente descendente/2;
5 represente um grafo com pelo menos 8 vértices e implemente alcanca/2;
6 identifique um caso em que a ordem das regras possa causar não terminação e

explique o motivo.

Estudo complementar - respostas de referência I

Estas são implementações possíveis; outras soluções equivalentes podem estar
corretas.

dobro(X, Y) :- Y is 2 * X.

primeiro(X, [X|_]).

ultimo(X, [X]).
ultimo(X, [_|T]) :- ultimo(X, T).

tamanho([], 0).
tamanho([_|T], N) :-

tamanho(T, N1),
N is N1 + 1.

concatena([], L, L).
concatena([H|T], L, [H|R]) :-

concatena(T, L, R).

Observação

Em dobro/2, X precisa estar instanciado para que 2*X seja avaliado por is.

Estudo complementar - respostas de referência II

A relação descendente pode reutilizar a relação ancestral:

descendente(X, Y) :- ancestral(Y, X).

Exemplo de grafo e alcançabilidade:

aresta(a,b). aresta(a,c). aresta(b,d). aresta(c,e).
aresta(d,f). aresta(e,f). aresta(f,g). aresta(g,h).

alcanca(X,Y) :- aresta(X,Y).
alcanca(X,Y) :- aresta(X,Z), alcanca(Z,Y).

Não terminação

Em um grafo com ciclo, por exemplo aresta(h,a), a regra recursiva acima pode
revisitar indefinidamente os mesmos vértices. Outro risco é colocar uma chamada
recursiva antes de uma condição que a restrinja suficientemente.

Estudo complementar - exemplos de testes

Testar um predicado significa verificar casos de sucesso, falha e, quando pertinente,
geração de respostas.

?- dobro(4, Y). % Y = 8
?- primeiro(X, [a,b,c]). % X = a
?- ultimo(X, [a,b,c]). % X = c
?- tamanho([a,b,c], N). % N = 3
?- concatena([a,b], [c,d], L). % L = [a,b,c,d]
?- descendente(eva, ana). % true
?- alcanca(a, h). % true
?- alcanca(h, a). % false, se nao houver ciclo

Boa prática

Para cada predicado, procure pelo menos um teste que deva ter sucesso e um que
deva falhar. Quando houver variáveis de saída, confira também as substituições
produzidas.

Desafio - conectando Prolog às aulas de busca

Para aprofundamento

No grafo criado no exercício anterior, implemente um predicado
caminho(Origem,Destino,Caminho) que devolva uma sequência de vértices
sem repetir estados.

1 mantenha uma lista de vértices já visitados;
2 impeça a expansão de um vértice já presente nessa lista;
3 explique qual papel essa lista desempenha em comparação com os algoritmos

de busca estudados anteriormente;
4 compare a ordem de exploração obtida com DFS.

Reflexão

O que mudou: o problema, o algoritmo de busca ou a forma como o algoritmo aparece
no programa?

Desafio - uma solução possível

Podemos manter uma lista de visitados e construir o caminho em ordem inversa
durante a busca:

caminho(Origem, Destino, Caminho) :-
caminho_aux(Origem, Destino, [Origem], Rev),
reverse(Rev, Caminho).

caminho_aux(Destino, Destino, Visitados, Visitados).

caminho_aux(Atual, Destino, Visitados, Caminho) :-
aresta(Atual, Proximo),
\+ member(Proximo, Visitados),
caminho_aux(Proximo, Destino,

[Proximo|Visitados], Caminho).

Papel de Visitados

A lista impede que um vértice seja expandido novamente no mesmo caminho. Ela
exerce o mesmo papel conceitual do controle de estados repetidos nos algoritmos de
busca.

Desafio - relação com DFS

aresta(Atual,Proximo) gera sucessores na ordem em que os fatos
aparecem na base;

o Prolog escolhe o primeiro sucessor disponível e chama recursivamente
caminho_aux/4;

portanto, aprofunda um ramo antes de testar os demais;

quando uma alternativa falha, o retrocesso retorna ao ponto de escolha anterior
e tenta outro sucessor;

esse comportamento corresponde à lógica de exploração em profundidade da
DFS.

Resposta à reflexão

O problema continua sendo encontrar um caminho. A estratégia de exploração
continua semelhante à DFS. O que muda é a forma de expressá-la: em Prolog, grande
parte do controle de busca aparece implicitamente no mecanismo de unificação,
recursão e retrocesso da linguagem.

Referências e ambiente de execução

SWI-Prolog Reference Manual e documentação oficial.

ISO/IEC 13211-1 - núcleo da linguagem Prolog.

Bratko, I. Prolog Programming for Artificial Intelligence.

Clocksin, W. F.; Mellish, C. S. Programming in Prolog.

Material-base anterior da disciplina, revisado e reorganizado para a sequência
atual.

Próxima aula: Lista 05

Como aplicar fatos, regras, unificação, recursão
e retrocesso na construção e análise de

programas Prolog?

Lista 05 - prática em Prolog

A próxima aula será destinada à resolução da Lista 05, consolidando o núcleo
estudado nesta aula e exigindo consulta ao material complementar quando necessário.

Dúvidas

Dúvida é o começo da sabedoria.

